

## Exercício 6 - Conectando-se a um PTT / IX

**Objetivo:** Conectar o Sistema Autônomo a um PTT / IX a partir do PoP. Serão abordados tópicos referentes às boas práticas de configuração e participação em um PTT / IX, além de trabalharmos com políticas de roteamento de entrada e de saída.

**Obs:** A rede do AS deve estar com OSPF e iBGP configurados e a sessão eBGP com a Operadora deve estar ativa.

A partir deste exercício o AS passará a ter uma segunda saída para redes externas, justificando a aquisição do ASN. Essa segunda saída será a conexão ao PTT feita através do roteador MikrotikBorda. A estrutura do PTT existente no laboratório é composta por dois roteadores, um *route server*, responsável por divulgar as rotas dos participantes, e um *looking glass*. Após a conclusão do exercício, todo tráfego de entrada e saída do AS fluirá preferencialmente pelo PTT.

1. O primeiro passo para estabelecer a sessão eBGP é adicionar os endereços IPv4 e IPv6 à interface `ether3` e adicionar essa interface as vlans 400 e 600, respectivamente, utilizada para a troca de tráfego multilateral no PTT.

- No roteador **MikrotikBorda** utilize os seguintes comandos:

```
/interface vlan
  add name=VLAN-ATM-IPV4-PTT-01 vlan-id=400 interface=ether3
  add name=VLAN-ATM-IPV6-PTT-01 vlan-id=600 interface=ether3

/ip address
  add address=102.111.0.XX netmask=255.255.255.0 \
interface=VLAN-ATM-IPV4-PTT-01 comment=EBGP-PTT-01
/ipv6 address
  add address=4d0c:111::XX/64 interface=VLAN-ATM-IPV6-PTT-01 \
comment=EBGP-PTT-01
```

2. O passo seguinte é estabelecer a sessão eBGP entre o roteador MikrotikBorda e o *looking glass*. Essa sessão não tem muitas restrições, sendo recomendado que se envie todas as informações de tabela de roteamento (*Full Routing*). Essa sessão deverá ser estabelecida com o endereço da interface física, fornecido pelo PTT.

- No roteador **MikrotikBorda** utilize os seguintes comandos:

```
/routing bgp peer
  add address-families=ip name=EBGP-IPV4-LG-PTT01 \
remote-address=102.111.0.253 remote-as=64511 \
nexthop-choice=force-self
  add address-families=ipv6 name=EBGP-IPV6-LG-PTT01 \
remote-address=4D0C:111::253 remote-as=64511 \
```

```
nexthop-choice=force-self
```

Verifique se a sessão foi estabelecida usando o seguinte comando no **MikrotikBorda**:

```
/routing bgp peer print detail status  
/routing bgp advertisements print peer=EBGP-IPV4-LG-PTT01  
/routing bgp advertisements print peer=EBGP-IPV6-LG-PTT01
```

3. O passo seguinte é configurar a sessão eBGP com o *route server*. Essa sessão deverá ser fechada também através do endereço da interface física, no entanto, ela deverá ser configurada com o comando *disable=yes* habilitado na configuração de vizinhança. Essa prática é recomendada para evitar que se estabeleça a sessão antes das políticas de roteamento serem aplicadas e com isso, evitar rotas indevidas acabem sendo anunciadas ao *route server*, que possui uma política de “derrubar” sessões eBGP que enviem para ele mais do que 5 prefixos.

**Obs: se não for configurado o peer com *disable=yes* o mikrotik irá divulgar todas as rotas conhecidas e o PTT irá bloquear sua conexão. Não pule essa etapa!**

- No roteador **MikrotikBorda** utilize os seguintes comandos:

```
/routing bgp peer  
  add address-families=ip name=EBGP-IPV4-RS-PTT01 \  
remote-address=102.111.0.254 remote-as=64501 \  
nexthop-choice=force-self disable=yes  
  add address-families=ipv6 name=EBGP-IPV6-RS-PTT01 \  
remote-address=4D0C:111::254 remote-as=64501 \  
nexthop-choice=force-self disable=yes
```

4. Com as configurações da sessão eBGP prontas, antes de retirar o comando *disable=yes*, configure as políticas de saída no roteador MikrotikBorda, aplique às rotas recebidas via PTT, respeitando os seguintes critérios:

- Políticas de saída:

- Anunciar apenas dois prefixos /33 e dois prefixos /23 pelo PTT.

- No roteador **MikrotikBorda** utilize os seguintes comandos:

```
/routing bgp network  
  add disabled=no network=102.XX.0.0/23 synchronize=no  
  add disabled=no network=102.XX.2.0/23 synchronize=no  
  add disabled=no network=4D0C:ABXX::/33 synchronize=no
```

```

add disabled=no network=4D0C:ABXX:8000::/33 synchronize=no
/routing filter
add action=accept chain=BGP-OUT-PTT01-IPV4 prefix=102.XX.0.0/23
add action=accept chain=BGP-OUT-PTT01-IPV4 prefix=102.XX.2.0/23
add action=discard chain=BGP-OUT-PTT01-IPV4 prefix=0.0.0.0/0 \
prefix-length=0-32
add action=accept chain=BGP-OUT-PTT01-IPV6 prefix=4D0C:ABXX::/33
add action=accept chain=BGP-OUT-PTT01-IPV6 \
prefix=4D0C:ABXX:8000::/33
add action=discard chain=BGP-OUT-PTT01-IPV6 prefix=::/0 \
prefix-length=0-128

/routing bgp peer
set EBG-IPV4-RS-PTT01 out-filter=BGP-OUT-PTT01-IPV4
set EBG-IPV6-RS-PTT01 out-filter=BGP-OUT-PTT01-IPV6

```

5. Agora, configure políticas de entrada no roteador MikrotikBorda, aplique as rotas recebidas via PTT, respeitando os seguintes critérios:

- Políticas de entrada:

- Aumentar o valor do atributo *Local Preference* para todas as rotas IPv4 e IPv6 aprendidas via PTT.
- Rejeitar o recebimento de prefixos que contenham rota *default*, prefixos do seu próprio AS e *bogons*, tanto IPv4 quanto para IPv6.

- No roteador **MikrotikBorda** utilize os seguintes comandos:

```

/routing filter
add action=discard chain=BGP-IN-PTT01-IPV4 prefix=102.XX.0.0/22 \
prefix-length=22-32
add action=discard chain=BGP-IN-PTT01-IPV4 prefix=102.120.0.0/16 \
prefix-length=16-32
add action=discard chain=BGP-IN-PTT01-IPV4 prefix=0.0.0.0/8 \
prefix-length=8-32
add action=discard chain=BGP-IN-PTT01-IPV4 prefix=10.0.0.0/8 \
prefix-length=8-32
add action=discard chain=BGP-IN-PTT01-IPV4 prefix=100.64.0.0/10 \
prefix-length=10-32
add action=discard chain=BGP-IN-PTT01-IPV4 prefix=127.0.0.0/8 \
prefix-length=8-32
add action=discard chain=BGP-IN-PTT01-IPV4 prefix=169.254.0.0/16 \
prefix-length=16-32
add action=discard chain=BGP-IN-PTT01-IPV4 prefix=172.16.0.0/12 \
prefix-length=12-32
add action=discard chain=BGP-IN-PTT01-IPV4 prefix=192.0.0.0/24 \
prefix-length=24-32
add action=discard chain=BGP-IN-PTT01-IPV4 prefix=192.0.2.0/24 \
prefix-length=24-32

```

```

    add action=discard chain=BGP-IN-PTT01-IPV4 prefix=192.168.0.0/16 \
prefix-length=16-32
    add action=discard chain=BGP-IN-PTT01-IPV4 prefix=198.18.0.0/15 \
prefix-length=15-32
    add action=discard chain=BGP-IN-PTT01-IPV4 prefix=198.51.100.0/24 \
prefix-length=24-32
    add action=discard chain=BGP-IN-PTT01-IPV4 prefix=203.0.113.0/24 \
prefix-length=24-32
    add action=discard chain=BGP-IN-PTT01-IPV4 prefix=224.0.0.0/4 \
prefix-length=4-32
    add action=discard chain=BGP-IN-PTT01-IPV4 prefix=0.0.0.0/0
    add action=accept chain=BGP-IN-PTT01-IPV4 set-bgp-local-pref=150

/routing filter
    add action=discard chain=BGP-IN-PTT01-IPV6 prefix=4D0C:ABXX::/32 \
prefix-length=32-128
    add action=discard chain=BGP-IN-PTT01-IPV6 prefix=4d0c:120::/32 \
prefix-length=32-128
    add action=discard chain=BGP-IN-PTT01-IPV6 prefix=2001:db8::/32 \
prefix-length=32-128
    add action=accept chain=BGP-IN-PTT01-IPV6 prefix=2001:500::/30 \
prefix-length=30-64 set-bgp-local-pref=150
    add action=accept chain=BGP-IN-PTT01-IPV6 prefix=2001::/32 \
prefix-length=32-64 set-bgp-local-pref=150
    add action=accept chain=BGP-IN-PTT01-IPV6 prefix=2001::/16 \
prefix-length=16-64 set-bgp-local-pref=150
    add action=accept chain=BGP-IN-PTT01-IPV6 prefix=2001:c00::/23 \
prefix-length=23-64 set-bgp-local-pref=150
    add action=accept chain=BGP-IN-PTT01-IPV6 prefix=2002::/16 \
prefix-length=16-64 set-bgp-local-pref=150
    add action=accept chain=BGP-IN-PTT01-IPV6 prefix=2003::/16 \
prefix-length=16-64 set-bgp-local-pref=150
    add action=accept chain=BGP-IN-PTT01-IPV6 prefix=2400::/12 \
prefix-length=12-64 set-bgp-local-pref=150
    add action=accept chain=BGP-IN-PTT01-IPV6 prefix=2600::/12 \
prefix-length=12-64 set-bgp-local-pref=150
    add action=accept chain=BGP-IN-PTT01-IPV6 prefix=2610::/23 \
prefix-length=23-64 set-bgp-local-pref=150
    add action=accept chain=BGP-IN-PTT01-IPV6 prefix=2620::/23 \
prefix-length=23-64 set-bgp-local-pref=150
    add action=accept chain=BGP-IN-PTT01-IPV6 prefix=2800::/12 \
prefix-length=12-64 set-bgp-local-pref=150
    add action=accept chain=BGP-IN-PTT01-IPV6 prefix=2a00::/12 \
prefix-length=12-64 set-bgp-local-pref=150
    add action=accept chain=BGP-IN-PTT01-IPV6 prefix=2801::/24 \
prefix-length=24-64 set-bgp-local-pref=150
    add action=accept chain=BGP-IN-PTT01-IPV6 prefix=2c00::/12 \
prefix-length=12-64 set-bgp-local-pref=150
    add action=accept chain=BGP-IN-PTT01-IPV6 prefix=4d0c::/16 \
prefix-length=16-64 set-bgp-local-pref=150

```

```
add action=discard chain=BGP-IN-PTT01-IPV6 prefix=::/0 \  
prefix-length=0-128  
  
/routing bgp peer  
set EBG-IPV4-RS-PTT01 in-filter=BGP-IN-PTT01-IPV4  
set EBG-IPV6-RS-PTT01 in-filter=BGP-IN-PTT01-IPV6
```

6. Outras boas práticas devem ser aplicadas para evitar o tráfego de pacotes indesejados, como:

- Desabilitar o MNDP (Mikrotik Neighbor Discovery Protocol)
- Desabilitar RA IPv6
- Filtrar pacotes cujo destino não seja para seu bloco IP (para evitar se tornar um trânsito dentro do PTT)

- No roteador **MikrotikBorda** utilize os seguintes comandos:

```
/ip neighbor discovery-settings  
set discover-interface-list=none  
  
/ipv6 nd print  
/ipv6 nd set 0 disabled=yes  
  
/ip firewall address-list  
add address=102.XX.0.0/22 list=FILTRO-PTT-V4  
  
/ip firewall filter  
add chain=forward in-interface=ether3 \  
src-address-list=FILTRO-PTT-V4  
add action=drop chain=forward in-interface=ether3  
  
/ipv6 firewall address-list  
add address=4D0C:ABXX::/32 list=FILTRO-PTT-V6  
  
/ipv6 firewall filter  
add chain=forward in-interface=ether3 \  
src-address-list=FILTRO-PTT-V6  
add action=drop chain=forward in-interface=ether3
```

Verifique se a sessão foi estabelecida usando o seguinte comando no MikrotikBorda:

7. Com políticas e filtros aplicados, habilite as sessões eBGP, IPv4 e IPv6. Para isso, execute os seguintes comandos no roteador **MikrotikBorda**:

```
/routing bgp peer set EBG-IPV4-RS-PTT01 disable=no  
/routing bgp peer set EBG-IPV6-RS-PTT01 disable=no
```

8. Verifique se essas sessões foram estabelecida usando os seguintes comando no MikrotikBorda:

```
/routing bgp peer print detail status
/routing bgp advertisements print peer=EBGP-IPV4-RS-PTT01
  PREFIX
  102.XX.0.0/23
  102.XX.2.0/23
/routing bgp advertisements print peer=EBGP-IPV6-RS-PTT01
  PREFIX
  4D0C:ABXX::/33
  4D0C:ABXX:8000::/33
```

- Quantos prefixos IPv4 e IPv6 estão sendo aprendidos via PTT?
- Há diferenças entre a quantidade de prefixos IPv4 e IPv6? Por quê?
- As políticas IPv4 e IPv6 de saída foram aplicadas corretamente?
- O valor do atributo *LocalPref* foi alterado corretamente em todos os prefixos aprendidos?

Acesse os outros roteadores do AS e verifique se os prefixos aprendidos via PTT estão sendo repassados corretamente para todo o ISP via iBGP com o valor do atributo *LocalPref* correto.

Veja também os prefixos anunciados no *looking glass* disponível no AS64511, do PTT. Para acessá-lo, utilize os seguintes comandos a partir do servidor **MikrotikBorda**:

```
/system telnet address=102.111.0.253 port=2605

> show ip bgp summary
> show bgp summary
> show ip bgp
> show bgp
> exit
```

Por quantos caminhos diferentes o AS64511 sabe chegar a rede de seu AS? As rotas para a rede IPv4 são similares às da rede IPv6?